

G01FAF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G01FAF returns the deviate associated with the given probability of the standard Normal distribution, via the routine name.

2 Specification

```

real FUNCTION G01FAF(TAIL, P, IFAIL)
  INTEGER          IFAIL
  real           P
  CHARACTER*1     TAIL

```

3 Description

The deviate, x_p associated with the lower tail probability, p , for the standard Normal distribution is defined as the solution to:

$$P(X \leq x_p) = p = \int_{-\infty}^{x_p} Z(X) dX$$

where

$$Z(X) = \frac{1}{\sqrt{2\pi}} e^{-X^2/2}, \quad -\infty < X < \infty.$$

The method used is an extension of that of Beasley and Springer [2]. p is first replaced by $q = p - 0.5$.

- (a) if $|q| \leq 0.3$, x_p is computed by a rational Chebyshev approximation

$$x_p = s \frac{A(s^2)}{B(s^2)}$$

where $s = \sqrt{2\pi} \cdot q$ and A, B are polynomials of degree 7.

- (b) if $0.3 < |q| \leq 0.42$, x_p is computed by a rational Chebyshev approximation

$$x_p = \text{sign } q \left(\frac{C(t)}{D(t)} \right)$$

where $t = |q| - 0.3$ and C, D are polynomials of degree 5.

- (c) if $|q| > 0.42$, x_p is computed as

$$x_p = \text{sign } q \left\{ \left(\frac{E(u)}{F(u)} \right) + u \right\}$$

where $u = \sqrt{-2 \times \log(\min(p, 1 - p))}$ and E, F are polynomials of degree 6.

For the upper tail probability $-x_p$ is returned while for the two tail probabilities the value x_{p^*} is returned where p^* is the required tail probability computed from the input value of p .

4 References

- [1] Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* Dover Publications (3rd Edition)
- [2] Beasley J D and Springer S G (1977) Algorithm AS111. The percentage points of the normal distribution *Appl. Statist.* **26** 118–120

[3] Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworths

5 Parameters

1: TAIL — CHARACTER*1

Input

On entry: indicates which tail the supplied probability represents.

If TAIL = 'L', the lower probability, i.e., $P(X \leq x_p)$.

If TAIL = 'U' the upper probability, i.e., $P(X \geq x_p)$.

If TAIL = 'S' the two tail (significance level) probability, i.e., $P(X \geq |x_p|) + P(X \leq -|x_p|)$.

If TAIL = 'C' the two tail (confidence interval) probability, i.e., $P(X \leq |x_p|) - P(X \leq -|x_p|)$.

Constraint: TAIL = 'L', 'U', 'S' or 'C'.

2: P — *real*

Input

On entry: the probability, p , from the standard Normal distribution as defined by TAIL.

Constraint: $0.0 < P < 1.0$.

3: IFAIL — INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors detected by the routine:

If on exit IFAIL \neq 0, then G01FAF returns 0.0.

IFAIL = 1

On entry, TAIL \neq 'L', 'U', 'S' or 'C'.

IFAIL = 2

On entry, $P \leq 0.0$,
or $P \geq 1.0$.

7 Accuracy

The accuracy is mainly limited by the *machine precision*.

8 Further Comments

None.

9 Example

Four values of TAIL and P are input and the deviates calculated and printed.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      G01FAF Example Program Text
*      Mark 15 Release. MAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
*      .. Local Scalars ..
      real            DEV, P
      INTEGER          I, IFAIL
      CHARACTER        TAIL
*      .. External Functions ..
      real            G01FAF
      EXTERNAL         G01FAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G01FAF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Tail      Probability      Deviate '
      WRITE (NOUT,*)
      DO 20 I = 1, 4
         READ (NIN,*) TAIL, P
         IFAIL = 0
*
         DEV = G01FAF(TAIL,P,IFAIL)
*
         WRITE (NOUT,99999) TAIL, P, DEV
      20 CONTINUE
      STOP
*
      99999 FORMAT (3X,A1,9X,F5.3,9X,F6.4)
      END

```

9.2 Program Data

```

G01FAF Example Program Data
'L' 0.975
'U' 0.025
'C' 0.95
'S' 0.05

```

9.3 Program Results

G01FAF Example Program Results

Tail	Probability	Deviate
L	0.975	1.9600
U	0.025	1.9600
C	0.950	1.9600
S	0.050	1.9600